[bookmark: _Hlk99919071]3GPP TSG-RAN5 Meeting #2022-TTCN email	R5s220485
Online, , 10th Dec 2021 - 31st Dec 2022
	CR-Form-v12.2

	CHANGE REQUEST

	

	
	34.229-3
	CR
	0803
	rev
	-
	Current version:
	16.5.0
	

	

	For HELP on using this form: comprehensive instructions can be found at 
http://www.3gpp.org/Change-Requests.

	



	Proposed change affects:
	UICC apps
	
	ME
	
	Radio Access Network
	
	Core Network
	



	

	Title:	
	Checking the absence of geolocation information in  INVITE request

	
	

	Source to WG:
	MCC TF160

	Source to TSG:
	R5

	
	

	Work item code:
	TEI9_Test
	
	Date:
	2022-04-04

	
	
	
	
	

	Category:
	F
	
	Release:
	Rel-16

	
	Use one of the following categories:
F  (correction)
A  (mirror corresponding to a change in an earlier 													release)
B  (addition of feature), 
C  (functional modification of feature)
D  (editorial modification)
Detailed explanations of the above categories can
be found in 3GPP TR 21.900.
	Use one of the following releases:
Rel-8	(Release 8)
Rel-9	(Release 9)
Rel-10	(Release 10)
Rel-11	(Release 11)
…
Rel-16	(Release 16)
Rel-17	(Release 17)
Rel-18	(Release 18)
Rel-19	(Release 19)

	
	

	Reason for change:
	There is no possibility to verify explicitly that geolocation information is not provided by the UE in INVITE request when such information is not available at the UE.

	
	

	Summary of change:
	Geolocation information was initialized as ‘omit’ in INVITE request. Depending upon geolocation information being available to the UE the expected parameters were adjusted.

	
	

	Consequences if not approved:
	A non-conformant UE may pass the test case

	
	

	Clauses affected:
	19.1.2, J.19.2. 

	
	

	
	Y
	N
	
	

	Other specs
	
	X
	 Other core specifications	
	

	affected:
	X
	
	 Test specifications
	TS 34.229-1

	(show related CRs)
	
	X
	 O&M Specifications
	

	
	

	Other comments:
	Prose CR is associated.

	
	

	This CR's revision history:
	


Page 1


[bookmark: _Toc100323325]Table of Contents
Table of Contents	2
1.	Correction to f_IMS_InviteRequest_MessageHeaderRX	2
2.	Correction to f_IMS_A_2_1_Invite_CommonChecks	6
3.	Correction to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional	9
4.	Correction to f_IMS_SRVCC_ReInvite_AnnexC24_Step1	10
5.	Correction to f_IMS_SRVCC_ReInvite_AnnexC31_Step1	11
6.	Correction to f_ECall_InitialInvite_AnnexC47	12
7.	Correction to f_IMS_INVITE_ReceiveRequest	14
8.	Correction to f_IMS_MOCallSetup_ReceiveINVITE_Optional	14

1. [bookmark: _Toc100323326]Correction to f_IMS_InviteRequest_MessageHeaderRX


	Function name
	f_IMS_InviteRequest_MessageHeaderRX

	Reason for change
	Geolocation and GeolocationRouting  not defaulted to omit in the template for Invite Request if condition A8 is false.

	Summary of change
	Geolocation and GeolocationRouting  defaulted to omit in the template for Invite Request if condition A8 is false.

	TTCN module
	IMS_Procedures_CallControl4G5G



	  function f_IMS_InviteRequest_MessageHeaderRX(template (present) SipUrl p_ExpectedCalleeUri,
                                               OptionTag_List p_OptionTagsForSupported := tsc_OptionTagList_100rel,
                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,
                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */
                                               boolean p_A8 := false,
                                               boolean p_A11 := false)
    runs on IMS_PTC return template (present) MessageHeader
  {
    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2
    var template (present) MessageHeader v_MessageHeader_Invite := cr_MessageHeader_Dummy;
    var template (present) RouteBody_List v_RouteBodyList;
    var IMS_ProtectedPorts_Type v_Protected;
    var template integer v_Port_us_Contact := *;
    var template integer v_Port_ps_Route := *;
    var template DisplayName v_DisplayName := *;
    var template (present) SipUrl v_SipUrlFrom := ?;    /* per default any SIP URI is accepted in the From header in the first place;
                                                           further checks may be applied in f_IMS_A_2_1_Invite_CommonChecks (depending on the conditions acc. to A.2.1 */
    var template (present) charstring v_Pcscf;
    var charstring v_MediaType := c_sdpApplication;
    var template (present) CallId v_CallId;
    var template (present) integer v_CSeqValue := ?;
    var template (present) From v_From;
    var template (present) To v_To;
    var template AcceptBody_List v_AcceptBodyList := {};
    var OptionTag_List v_OptionTag_List := p_OptionTagsForSupported;
    var boolean v_GRUUsInSIP := pc_IMS_GRUUsInSIP;
    var boolean v_A4 := (p_Context == A_2_1_A4);    // INVITE creating a dialog    @sic R5-166280 sic@
    var boolean v_A5 := (p_Context == A_2_1_A5);    // re-INVITE within a dialog   @sic R5-166280 sic@
    var boolean v_A6 := (p_Context == A_2_1_A6);    // emergency session in case of no registration          @sic R5-166280 sic@
    var boolean v_A19 := (p_Context == A_2_1_A19);  // emergency session in case of no registration (GIBA)   @sic R5-174545 sic@
    var boolean v_A26 := pc_IMS_Session_Timer;      // UE supports Session Timer   @sic R5-192321 sic@
    
    if (pc_IMS_CS_PS_SRVCCMidCall or pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@
      v_OptionTag_List := v_OptionTag_List & tsc_OptionTagList_norefersub;
    }
    if (v_A4 and v_A26) {                           /* @sic R5-192321 sic@
                                                       NOTE: A4, A6, A7 and A19 are mutual exclusive; use of A20, A21 (eCall) is unclear */
      v_OptionTag_List := v_OptionTag_List & tsc_OptionTagList_timer;  /* @sic R5-173024: support of session timer acc. RFC 4028  sic@
                                                                          @sic R5-180230: end of grace period sic@ */
    }
    
    if (v_A6 or v_A19) {  // INVITE for creating an emergency session in case of no registration
      // Route, Contact: ports to be used
      v_Port_us_Contact := ?;
      v_Port_ps_Route := tsc_IMS_PortNumber_5060;
      if (not v_A19) {                                    /* @sic R5-191118 sic@ */
        // From: Display
        v_DisplayName := cr_DisplayName("Anonymous");     /* @sic R5s140005, R5-140901, R5w140111 sic@ */
      }
      v_GRUUsInSIP := false;     /* @sic R5s160366: no GRUU in case of no registration sic@ */
    }
    else {
      // Route, Contact: ports to be used
      if (not v_IsGiba or (p_Context == A_2_1_A7)) {
        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();
        v_Port_us_Contact := v_Protected.Port_us;
        v_Port_ps_Route := v_Protected.Port_ps;
      } else {  /* @sic R5s160366: GIBA sic@ */
        v_Port_us_Contact := ?;
        v_Port_ps_Route := tsc_IMS_PortNumber_5060 ifpresent;
      }
    }

    if (v_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */
      v_RouteBodyList := f_IMS_Request_RouteSet_RouteBodyList_RX();
      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();
      v_From := f_IMS_RequestInDialog_FromHeaderRX();
      v_To := f_IMS_RequestInDialog_ToHeaderRX();
      v_CSeqValue := f_IMS_RequestInDialog_CSeqValueRX();     /* @sic R5s140235, R5s140385 change 3 MCC160 implementation sic@ */
    }
    else {                                       /* normal case: no re-invite */
      // route body acc. to conditions in A.2.1
      v_Pcscf := ?;                              /* @sic R5s140934 change 6: P-CSCF address is checked later on (f_IMS_A_2_1_Invite_CommonChecks -> f_IMS_MessageHeader_Request_CheckRoute) sic@ */
      v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(v_Pcscf, v_Port_ps_Route));  /* @sic R5s140873: v_Pcscf;  R5s140934 change 6 sic@
                                                                                                  @sic R5-155174: "loose routing" parameter mandatory for any kind of call (i.e. emergency call too, in contrast to R5s150817) sic@ */
      if (not match(p_Context, (A_2_1_A6, A_2_1_A7, A_2_1_A19))) {  // => no emergency call
        v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(tsc_IMS_Scscf));
      }
      if (p_A8) {
        v_MessageHeader_Invite.geolocation := cr_GeolocationAny;
        v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@
        v_MediaType := tsc_MimeType_MultipartMixed;
      }
      v_CallId := cr_CallId(?);
      v_From := cr_From(v_DisplayName, v_SipUrlFrom);
      v_To := cr_ToDef(p_ExpectedCalleeUri);
    }
    
    if (p_A8) {
      v_MessageHeader_Invite.geolocation := cr_GeolocationAny;
      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@
      v_MediaType := tsc_MimeType_MultipartMixed;
    } else {
      v_MessageHeader_Invite.geolocation := omit;
      v_MessageHeader_Invite.geolocationRouting := omit;
    }
    
    v_MessageHeader_Invite.via             := f_IMS_SipRequest_ViaRX();                               /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address
                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */
    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);
    v_MessageHeader_Invite.fromField       := v_From;                                                 /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */
    v_MessageHeader_Invite.toField         := v_To;                                                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */
    v_MessageHeader_Invite.callId          := v_CallId;                                               /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */
    v_MessageHeader_Invite.cSeq            := cr_CseqDef(v_CSeqValue, "INVITE");                      /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */
    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);
    v_MessageHeader_Invite.contact         := cr_Contact(f_Contact_SipUri_HostPortRX(?, v_Port_us_Contact, v_GRUUsInSIP));  /* additional checks in f_IMS_A_2_1_Invite_CommonChecks:
                                                                                                                               check of FQDN or IP address, check of feature-param for A3, A6, A10, A12
                                                                                                                               @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@
                                                                                                                               @sic R5s160366: v_GRUUsInSIP sic@ */
    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);
    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;
    
    // additional headers for specific conditions:
    if (v_IsGiba or v_A6 or v_A19) {
      v_MessageHeader_Invite.securityVerify  := omit;
      if (v_A6 or v_A19) {
        v_MessageHeader_Invite.require       := omit;
        v_MessageHeader_Invite.proxyRequire  := omit;
      }
    } else {   // -> A1 or A7
      v_MessageHeader_Invite.securityVerify  := f_IMS_PTC_BuildSecurityVerifyHeader();
      if (not v_A5) {  /* @sic R5s140017 change 9.3 sic@ */
        v_MessageHeader_Invite.require       := cr_RequireSecAgree;
        v_MessageHeader_Invite.proxyRequire  := cr_ProxyRequireSecAgree;
      }
    }

    if (not v_IsGiba) {
      v_MessageHeader_Invite.pAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());
    }

    if (v_A4) {  /* @sic R5s140017 change 9.3 sic@ */
      v_AcceptBodyList := {cr_AcceptBody(c_sdpApplication), cr_AcceptBody(c_ims3gppApplication)};
    }
    if (pc_IMS_CS_PS_SRVCCMidCall) { // @sic R5-145746 sic@
      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.mid-call+xml"); // add another element in the array
    }
    if (pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@
      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.state-and-event-info+xml"); // add another element in the array
      v_MessageHeader_Invite.recvInfo := cr_RecvInfo_Def(tsc_IMS_InfoPackageType_3gpp);  // @sic R5-145746, R5s150704,  sic@
    }
    if (lengthof(v_AcceptBodyList) > 0) {
      if (v_A5) { // @sic R5s150971 header optional if A5 sic@
        v_MessageHeader_Invite.accept := cr_Accept(superset(all from v_AcceptBodyList)) ifpresent;
      } else {
        v_MessageHeader_Invite.accept := cr_Accept(superset(all from v_AcceptBodyList));
      }
    }
    
    if (p_A8) {
      v_MessageHeader_Invite.geolocation     := cr_GeolocationAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks
      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks @sic R5-133151 sic@
    }

    if (p_Context == A_2_1_A7) {
      v_MessageHeader_Invite.pPreferredID := cr_PPreferredID(?);    // Checked in f_IMS_A_2_1_Invite_CommonChecks
    }
    if (p_A3 and v_A4) {                          /* @sic R5s140263 change 5, modified by R5-142949 (-> A4 only) sic@ */
      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_FeatureParamValue_3gpp_icsi_ref);   /* @sic R5s140565 change 2 sic@ */
    }
    if ((p_A3 and v_A4) or p_A11) {               /* @sic R5-142949 sic@ */
      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)
    }

    if (pc_EarlyMedia and not v_A5) {                                             /* @sic R5-162934: condition A16 for A.2.1  sic@
                                                                                     @sic R5-166280: "not A5" sic@ */
      
      v_MessageHeader_Invite.pEarlyMedia := cr_PEarlyMedia({"supported"});
    }
    return v_MessageHeader_Invite;
  }



2. [bookmark: _Toc100323327]Correction to f_IMS_A_2_1_Invite_CommonChecks

	Function name
	f_IMS_A_2_1_Invite_CommonChecks

	Reason for change
	Geolocation and GeolocationRouting  not needed to be checked any more in this function because of the changes in template for Invite Request

	Summary of change
	Geolocation and GeolocationRouting  not checked any more in this function because of the changes in template for Invite Request

	TTCN module
	IMS_Procedures_CallControl4G5G



	  function f_IMS_A_2_1_Invite_CommonChecks(IMS_DATA_REQ p_IMS_INVITE_REQ,
                                           INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,
                                           boolean p_A3 := false,
                                           boolean p_A8 := false,
                                           boolean p_A11 := false,
                                           boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert,
                                           boolean p_OriginatingIdentificationRestriction := false,
                                           boolean p_A18 := pc_BSRVCC) runs on IMS_PTC
  {    var MessageHeader v_MessageHeader := p_IMS_INVITE_REQ.Request.Invite.msgHeader;
    var SemicolonParam_List v_ContactParams := {};
    var SemicolonParam_List v_AcceptContactParams := {};
    var InternetProtocol_Type v_Protocol := p_IMS_INVITE_REQ.RoutingInfo.Protocol;
    var SipUrl v_SipUrlFrom := f_Addr_Union_GetSipUrl(v_MessageHeader.fromField.addressField);
    var SipUrl v_SipUrlPreferred;
    var template SipUrl v_ExpectedEmergencyUrl := f_IMS_PTC_ImsInfo_GetSipUriRX(emergencyId, *);
    var PPreferredIDValue_List v_PreferredIDValueList;
    var CharStringList_Type v_PreferredPublicUserIdList;
    var boolean v_EmergencyCallWithoutRegistration := (p_Context == A_2_1_A6);
    var boolean v_AllowFQDN := not v_EmergencyCallWithoutRegistration;  // for A6 there is no FQDN for the UE
    var boolean v_CheckRport := v_EmergencyCallWithoutRegistration;
    var boolean v_A10 := pc_IMS_Video_FeatureTag;
    var boolean v_A12 := p_SrvccAlertingSupported and (p_Context != A_2_1_A5);  /* @sic R5s140370 change 7 sic@ */
    var boolean v_A29 := pc_NG114_v1_0_mmtel;
    var boolean v_A30 := pc_NG114_v1_0_audio;
    var boolean v_A31 := pc_NG114_v1_0_video;
    var integer i;
    var template (present) GenValue v_SipInstanceValue;
    var template integer v_Port_us_Via := *;

    if ((v_EmergencyCallWithoutRegistration and (v_Protocol==udp))) {  /* @sic R5-172412 sic@ */
      v_Port_us_Via := tsc_IMS_PortNumber_5060;
    }

    if (ispresent(v_MessageHeader.contact.contactBody.contactAddresses[0].contactParams)) {
      v_ContactParams := v_MessageHeader.contact.contactBody.contactAddresses[0].contactParams;
    }
    if (ispresent(v_MessageHeader.acceptContact.acValues[0].acRcParams)) {
      v_AcceptContactParams := v_MessageHeader.acceptContact.acValues[0].acRcParams;
    }

    // ****  Common checks ****
    f_IMS_MessageHeader_Request_CheckVia(v_MessageHeader, v_Protocol, v_AllowFQDN, v_CheckRport, v_Port_us_Via);   /* check whether via contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)
                                                                                                                      @sic R5-172412: v_Port_us_Via sic@ */
    f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader, v_EmergencyCallWithoutRegistration);      /* check whether contact contains "SIP URI with IP address or FQDN" (see @sic R5s120727 sic@)
                                                                                                       @sic R5s160366: v_EmergencyCallWithoutRegistration sic@ */
    if (p_Context != A_2_1_A5) {
      f_IMS_MessageHeader_CheckTo(v_MessageHeader, noToTag);                                        /* check that there is no To tag */
      f_IMS_MessageHeader_Request_CheckRoute(v_MessageHeader);                                      /* @sic R5s140934 change 7 sic@ */
    }

    // ****  Check URIs in From header, P-Preferred-Identity header and further headers ****
    select (p_Context) {
      case (A_2_1_A4) {
        // Check Call-Id
        if (match(v_MessageHeader.callId.callid, f_IMS_PTC_ImsInfo_GetCallIdRegistration())) {
          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A4: invalid 'CallId' in INVITE (has to be different than in previous REGISTER)");
        }
        // Additional checks regarding SIP URI in the From header @sic R5-130750 change 3 sic@
        if (not p_OriginatingIdentificationRestriction) { /* @sic R5-160975, R5s160687: p_OriginatingIdentificationRestriction sic@ */
          if (not f_IMS_PublicUserId_Match(f_IMS_PTC_ImsInfo_GetAllIMPUs(), v_SipUrlFrom)) {
            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid 'From'");
          }
          if (ispresent(v_MessageHeader.pPreferredID.pPreferredIDValueList)) {
            v_PreferredIDValueList := v_MessageHeader.pPreferredID.pPreferredIDValueList;
            for (i:=0; i < lengthof(v_PreferredIDValueList); i := i+1) {
              v_PreferredPublicUserIdList[i] := f_SIP_SipUrl_GetPublicUserIdentity(f_Addr_Union_GetSipUrl(v_PreferredIDValueList[i]));
            }
            if (not f_IMS_PublicUserId_Match(v_PreferredPublicUserIdList, v_SipUrlFrom)) {
              f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A4: invalid 'From': does not match with P-Preferred-Identity");
            }
          }
        }
      }
      case (A_2_1_A5) {
        /* @sic R5s140017 MCC160 implementation: no further checks for A5 are needed in this function anymore sic@ */
      }
      case (A_2_1_A6) {
        v_SipInstanceValue := cr_GenValueQuoted(f_SIP_InstanceId()); /* @sic R5s130756 change 11; R5w140111: GenValue; R5s141212  sic@ */
        if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParam("+sip.instance", v_SipInstanceValue))) {
          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "IMEI code of the UE, coded as specified in draft-montemurro-gsma-imei-urn, not matching");
        }
      }
      case (A_2_1_A7) {
        if (not match(v_SipUrlFrom, v_ExpectedEmergencyUrl)) {
          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A7: invalid 'From'");
        }
        v_SipUrlPreferred := f_Addr_Union_GetSipUrl(v_MessageHeader.pPreferredID.pPreferredIDValueList[0]);
        if (not match(v_SipUrlPreferred, v_ExpectedEmergencyUrl)) {
          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A7: invalid 'P-Preferred-Identity'");
        }
      }
      case (A_2_1_A19)  { // @sic R5-174545, Change 2 R5s180463 sic@
        // Via, Route and Contact header are checked in function f_IMS_InviteRequest_MessageHeaderRX
        if (not match(v_SipUrlFrom, v_ExpectedEmergencyUrl)) {          // @sic R5-191118 sic@
          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A19: invalid 'From'");
        }
      }
      case else {
        FatalError(__FILE__, __LINE__, "unknown context");
      }
    }

    // ****  Check further Contact feature-param acc. to further conditions of A.2.1:  ****
    //
    if (p_A3 or v_A29) {                        // A.1.2 condition A3: MTSI  @sic R5-211418: v_A29 sic@
      if (not match(v_ContactParams, cr_FeatureParamList_3gpp_icsi_ref)) {            // @sic R5s130738: string needs to be quoted; R5-142247 sic@
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A3: Missing feature-param in Contact header");
      }
    }
    if (p_A3 and (p_Context == A_2_1_A4)) {     // @sic R5-142949 sic@
      if (not match(v_AcceptContactParams, cr_FeatureParamList_3gpp_icsi_ref)) {    // @sic R5s130738: string needs to be quoted; R5-142247 sic@
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A3: Missing feature-param in AcceptContact header");
      }
    }
    if (p_A8) {                                 // A.1.2 condition A8: check geolocation and geolocationRouting @sic R5-133151 sic@
      /* NOTE: v_MessageHeader.geolocation and geolocationRouting are checked already by template returned by f_IMS_InviteRequest_MessageHeaderRX:
       *       geolocation and geolocationRouting to be present. @sic R5-133151 sic@ */
      if (not match(v_MessageHeader.geolocationRouting.state, GEOLOCATION_ROUTING_YES_E)) {
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A8: Wrong geolocation-routing");
      }
    }
    if (v_A10 or v_A31) {    /* A.1.2 condition A10/A31: UE supports video feature tag   @sic R5-211418: v_A31 sic@
                                A.1.2 condition A11: INVITE for creating a video call
                                @sic R5-153766: "v_A10 and p_A11 and (p_Context != A_2_1_A6) and (p_Context != A_2_1_A7)" sic@
                                @sic R5-160553: "v_A10 and p_A11" sic@
                                @sic R5-176371: "v_A10" for ContactParams, "v_A10 and p_A11" for AcceptContactParams sic@ */
      if (not match(v_ContactParams, cr_VideoInContactParams)) {
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A10, A11: Missing feature-param in Contact header");
      }
      if (p_A11 and not match(v_AcceptContactParams, cr_VideoInContactParams)) {  /* @sic R5-176371 sic@ */
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A10, A11: Missing feature-param in Accept-Contact header");
      }
    }
    if ((v_A30 or f_IMS_PTC_AudioFeatureTagIsRequired()) and not match(v_ContactParams, cr_AudioInContactParams)) {    // check 'feature-param'  @sic R5-165943, R5-170564 sic@ @sic R5-173021 sic@  @sic R5-177127 sic@  @sic R5-211418: v_A30 sic@
      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "missing audio feature-param in Contact header");
    }
    if (v_A12) {                                                // A.1.2 condition A12: INVITE for creating a voice or video call and UE supports g.3gpp.srvcc-alerting media feature tag
      if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParamNoValue(tsc_IMS_FeatureCap_SrvccAlerting))) { /* @sic R5s140370 change 7: tsc_IMS_FeatureCap_SrvccAlerting sic@ */
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A12: Missing feature-param");
      }
    }
    if (p_A18 and (p_Context != A_2_1_A5)) {                    // A.1.2 condition A18: UE indicates g.3gpp.ps2cs-srvcc-orig-pre-alerting media feature tag in INVITE request @sic R5-171486 sic@ @sic R5-182793 sic@
      if (not match(v_ContactParams, cr_SemicolonParam_List_OneSpecificParamNoValue(tsc_IMS_FeatureCap_SrvccPreAlerting))) {
        f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "A.2.1 condition A18: Missing feature-param");
      }
    }
  }



3. [bookmark: _Toc100323328]Correction to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional

	Function name
	f_IMS_INVITE_ReceiveRequest_A_2_1_Optional

	Reason for change
	Geolocation and GeolocationRouting  not needed to be checked any more in this function because of the changes in template for Invite Request. 
Parameter A8 becomes superfluous.

	Summary of change
	Geolocation and GeolocationRouting  not checked any more in this function because of the changes in template for Invite Request. 
Parameter A8 becomes superfluous.

	TTCN module
	IMS_Procedures_CallControl4G5G



	  function f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(INVITE_A_2_1_Context_Type p_Context,
                                                      template (present) SipUrl p_ExpectedCalleeUri,
                                                      template (present) MessageHeader p_MessageHeader_Invite,
                                                      template (omit) float p_WaitDuration := omit,
                                                      boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */
                                                      boolean p_A8 := false,
                                                      boolean p_A11 := false,
                                                      boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert,
                                                      boolean p_OriginatingIdentificationRestriction := false)
    runs on IMS_PTC return template (omit) IMS_DATA_REQ
  {
    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;
    var boolean v_UseMIME := (ispresent(p_MessageHeader_Invite.contentType) and match(tsc_MimeType_MultipartMixed, p_MessageHeader_Invite.contentType.mTypeSubtype));
    var template (present) MessageBody v_MessageBody := f_SIP_MessageBodySDP_RX(v_UseMIME);  /* @sic R5w140112 sic@ @sic R5s210009: v_UseMIME instead of p_A8 sic@ */
    var boolean v_IsReInvite := (p_Context == A_2_1_A5);
    var boolean v_IsEmergencyCallWithoutRegistration := (p_Context == A_2_1_A6);

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_Optional(p_ExpectedCalleeUri, p_MessageHeader_Invite, v_MessageBody, p_WaitDuration, v_IsReInvite, v_IsEmergencyCallWithoutRegistration);    // @sic R5s190882 sic@
    if (isvalue(v_IMS_DATA_REQ)) {
      f_IMS_A_2_1_Invite_CommonChecks(valueof(v_IMS_DATA_REQ), p_Context, p_A3, p_A8, p_A11, p_SrvccAlertingSupported, p_OriginatingIdentificationRestriction);  /* @sic R5-160975: p_OriginatingIdentificationRestriction sic@ */
    }
    return v_IMS_DATA_REQ;
  }



4. [bookmark: _Toc100323329]Correction to f_IMS_SRVCC_ReInvite_AnnexC24_Step1

	Function name
	f_IMS_SRVCC_ReInvite_AnnexC24_Step1

	Reason for change
	Parameter A8 no longer needed to be passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	Summary of change
	Parameter A8 no longer passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	TTCN module
	IMS_CommonSRVCC



	  function f_IMS_SRVCC_ReInvite_AnnexC24_Step1(charstring p_OtherCalleeUri,
                                               template (omit) float p_WaitDuration := omit,
                                               boolean p_GeolocationInfoAvailable := false)
    runs on IMS_PTC return template (omit) IMS_InviteRequestWithSdp_Type
  { 
    var template (present) SDP_Message v_SDP_MessageRX := f_IMS_BuildSDP_AnnexC24_Step1();
    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;
    var IMS_DATA_REQ v_IMS_DATA_REQ_Val;
    var SDP_Message v_SDP_Message;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A5;
    var template (present) SipUrl v_CalleeSipUrl := f_SIP_BuildSipUri_RX(p_OtherCalleeUri, *);  /* @sic R5-151957 sic@ */
    var template (present) MessageHeader v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_CalleeSipUrl, -, v_Context, -, p_GeolocationInfoAvailable);    // @sic R5s190882 sic@
    
    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_CalleeSipUrl, v_MessageHeader_Invite, p_WaitDuration, -, p_GeolocationInfoAvailable);
    if (isvalue(v_IMS_DATA_REQ)) {
      v_IMS_DATA_REQ_Val := valueof(v_IMS_DATA_REQ);
      v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_IMS_DATA_REQ_Val.Request.Invite.messageBody, v_SDP_MessageRX, p_GeolocationInfoAvailable);
    
      return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ_Val, v_SDP_Message);
    }
    return omit;
  }



5. [bookmark: _Toc100323330]Correction to f_IMS_SRVCC_ReInvite_AnnexC31_Step1

	Function name
	f_IMS_SRVCC_ReInvite_AnnexC31_Step1

	Reason for change
	Parameter A8 no longer needed to be passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	Summary of change
	Parameter A8 no longer passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	TTCN module
	IMS_CommonSRVCC



	  function f_IMS_SRVCC_ReInvite_AnnexC31_Step1(charstring p_OtherCalleeUri,
                                               boolean p_GeolocationInfoAvailable := false,
                                               template (omit) float p_WaitDuration := omit)  // @sic R5-174016 sic@
    runs on IMS_PTC return template (omit) IMS_InviteRequestWithSdp_Type
  {
    var template (omit) float v_WaitDuration := p_WaitDuration;   //  @sic R5-174016 sic@
    var template (present) SDP_Message v_SDP_MessageRX := f_IMS_BuildSDP_AnnexC31_Step1();
    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;
    var IMS_DATA_REQ v_IMS_DATA_REQ_Val;
    var SDP_Message v_SDP_Message;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A5;
    var template (present) SipUrl v_CalleeSipUrl := f_SIP_BuildSipUri_RX(p_OtherCalleeUri, *);    /* @sic R5s150204 sic@ */
    var template (present) MessageHeader v_MessageHeader_Invite;

    v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_CalleeSipUrl, -, v_Context, -, p_GeolocationInfoAvailable);    // @sic R5s190882 sic@
    v_MessageHeader_Invite.reason := f_IMS_SIP_Reason487_RX();     // @sic R5s140370 change 3 - MCC160 implementation sic@

    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_CalleeSipUrl, v_MessageHeader_Invite, v_WaitDuration, -, p_GeolocationInfoAvailable);
    if (isvalue(v_IMS_DATA_REQ)) { // @sic R5-174016 sic@
      v_IMS_DATA_REQ_Val := valueof(v_IMS_DATA_REQ);
      v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_IMS_DATA_REQ_Val.Request.Invite.messageBody, v_SDP_MessageRX, p_GeolocationInfoAvailable);

      return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ_Val, v_SDP_Message);
    }
    return omit;
  }



6. [bookmark: _Toc100323331]Correction to f_ECall_InitialInvite_AnnexC47

	Function name
	f_ECall_InitialInvite_AnnexC47

	Reason for change
	Parameter A8 no longer needed to be passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	Summary of change
	Parameter A8 no longer passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	TTCN module
	IMS_eCall_Functions



	  function f_ECall_InitialInvite_AnnexC47(ECall_Type p_ECallType) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var MIME_Message v_MIME_Message;
    var MIME_Part v_MIME_Part;
    var SDP_Message v_SdpOffer;
    var template (present) SDP_Message v_SDP_Message_Step1;
    var template (present) SipUrl v_ECallPSAP_Rx := f_SipUrl_eCall_RX(p_ECallType);
    var template (present) MessageHeader v_MessageHeader;
    var template (present) AcceptBody_List v_AcceptBodyList := { cr_AcceptBody(tsc_MimeType_EmergencyCallDataControlXML), * };    // condition A20, A21
    var template (present) SemicolonParam_List v_DispositionParams := { cr_GenericParam("handling", "optional"), * };             // condition A20, A21
    var boolean v_A8 := pc_IMS_Geolocation;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A7;
    var charstring v_CidUrl;
	var ECall_Type v_ECallType := p_ECallType;
    
    // Step 1. Receive INVITE for an eCall
    v_MessageHeader := f_IMS_InviteRequest_MessageHeaderRX(v_ECallPSAP_Rx, -, v_Context, -, v_A8);
	if ((v_ECallType == MANUAL) or (v_ECallType == AUTOMATIC)) { // @sic R5-220971 sic@
      v_MessageHeader.callInfo := cr_CallInfo(-, "EmergencyCallData.eCall.MSD");     // condition A20, A21 @sic R5-210324: "EmergencyCallData" instead of "emergencyCallData" sic@
      v_MessageHeader.recvInfo := cr_RecvInfo_Def("EmergencyCallData.eCall.MSD");    // condition A20, A21
      v_MessageHeader.accept := cr_Accept(v_AcceptBodyList);                         // condition A20, A21
	}
    v_MessageHeader.contentType := cr_ContentTypeMIME;                             // condition A20, A21, A8, A25 @sic R5-217960 sic@
    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ECallPSAP_Rx, v_MessageHeader, -, -, v_A8));
    v_MIME_Message := v_IMS_DATA_REQ.Request.Invite.messageBody.mimeMessageBody;

    // Check PIDF
    if (v_A8) {
      v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, tsc_MimeType_Pidf);
      f_IMS_DecodeAndCheckGeolocation(v_MIME_Part.body.xmlBody);
    }
    // Check SDP
    v_SDP_Message_Step1 := f_IMS_BuildSDP_EmergencyCallOffer_RX();   // Build RX SDP template using codecs AMR and AMR_WB
    v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, c_sdpApplication);
    v_SdpOffer := f_IMS_SIP_DecodeMatchAndCheckSDP(v_MIME_Part.body.sdpMessageBody, v_SDP_Message_Step1);

    // Check eCall MSD
    v_CidUrl := v_IMS_DATA_REQ.Request.Invite.msgHeader.callInfo.callInfoBody[0].url;
    f_IMS_ECall_MIME_DecodeAndCheckEmergencyCallData(v_MIME_Message, tsc_MimeType_EmergencyCallDataMSD, v_CidUrl, v_DispositionParams);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ, v_SdpOffer);
  }



7. [bookmark: _Toc100323332]Correction to f_IMS_INVITE_ReceiveRequest

	Function name
	f_IMS_INVITE_ReceiveRequest

	Reason for change
	Parameter A8 no longer needed to be passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	Summary of change
	Parameter A8 no longer passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	TTCN module
	IMS_Procedures_CallControl4G5G



	  function f_IMS_INVITE_ReceiveRequest(INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,
                                       OptionTag_List p_OptionTagsForSupported := tsc_OptionTagList_100rel,
                                       template SipUrl p_OtherCalleeUri := omit,
                                       boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */
                                       boolean p_A8 := false,
                                       boolean p_A11 := false,
                                       boolean p_SrvccAlertingSupported := pc_IMS_SRVCCAlert)
    runs on IMS_PTC return IMS_DATA_REQ
  {
    var template (omit) float v_WaitDuration := omit;   // no wait duration
    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;
    var template (present) SipUrl v_ExpectedCalleeUri := f_IMS_INVITE_CalleeUriRX(p_OtherCalleeUri, p_Context);
    var template (present) MessageHeader v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_ExpectedCalleeUri, p_OptionTagsForSupported, p_Context, p_A3, p_A8, p_A11);    // @sic R5s190882 sic@
    
    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(p_Context, v_ExpectedCalleeUri, v_MessageHeader_Invite, v_WaitDuration, p_A3, p_A8, p_A11, p_SrvccAlertingSupported);
    return valueof(v_IMS_DATA_REQ);                     // as we don't have any WaitDuration f_IMS_INVITE_ReceiveRequest_A_2_1_Optional returns a value (omit could only occur when a timer expires)
  }




8. [bookmark: _Toc100323333]Correction to f_IMS_MOCallSetup_ReceiveINVITE_Optional

	Function name
	f_IMS_MOCallSetup_ReceiveINVITE_Optional

	Reason for change
	Parameter A8 no longer needed to be passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	Summary of change
	Parameter A8 no longer passed to f_IMS_INVITE_ReceiveRequest_A_2_1_Optional  as it has become superfluous.

	TTCN module
	IMS_Procedures_CallControl4G5G



	  function f_IMS_MOCallSetup_ReceiveINVITE_Optional(template (omit) float p_WaitDuration := omit,
                                                    template SipUrl p_OtherCalleeUri := omit,
                                                    template SDP_Message p_SDP_Message := omit,
                                                    IMS_SDP_AudioCodecList_Type p_AudioCodecList,
                                                    boolean p_PreconditionSupported := true,
                                                    boolean p_OIR := false,
                                                    boolean p_GeolocationInfoAvailable := false) runs on IMS_PTC return template (omit) IMS_InviteRequestWithSdp_Type
  { 
    var OptionTag_List v_OptionTagsForSupported := f_IMS_Invite_OptionTagsSupported(p_PreconditionSupported);  /* @sic R5s180197 sic@ */
    var template (omit) IMS_DATA_REQ v_IMS_DATA_REQ;
    var template (omit) IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp := omit;
    var template (present) SDP_Message v_SDP_MessageTemplate;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A4;
    var template (present) SipUrl v_ExpectedCalleeUri := f_IMS_INVITE_CalleeUriRX(p_OtherCalleeUri, v_Context);
    var template (present) MessageHeader v_MessageHeader_Invite := f_IMS_InviteRequest_MessageHeaderRX(v_ExpectedCalleeUri, v_OptionTagsForSupported, v_Context, -, p_GeolocationInfoAvailable);  // @sic R5s190882 sic@
    var IMS_DATA_REQ v_IMS_DATA_REQ_Value;
    var SDP_Message v_SDP_Message;
    var INVITE_Request v_InviteRequest;
    var boolean v_CheckForbiddenCodecParameters;                /* @sic R5-180231 sic@ */

    if (ispresent(p_SDP_Message)) {                             /* specific SDP message content @sic R5s180197 sic@ */
      v_SDP_MessageTemplate := p_SDP_Message;
      v_CheckForbiddenCodecParameters := false;                 /* no automatic check for forbidden parameters in case of specific (i.e. non C.21(abcd)) SDP messages */
    }
    else {                                                      /* SDP message content according to C.21(abcd) */
      v_SDP_MessageTemplate := f_IMS_BuildSDP_InitialOfferAudio_RX(p_PreconditionSupported, p_AudioCodecList); /* @sic R5s150704 - additional changes sic@ @sic R5-172931 sic@ */
      v_CheckForbiddenCodecParameters := true;                  /* automatic check for forbidden parameters for Cat-M1 (C.21d), EVS (C.44), WLAN (C.21a) and normal cases (C.21);
                                                                   @sic R5-180231: grace period, R5-192309: grace period for normal cases expired sic@ */
    }
    // Step 2. Receive INVITE Request
    v_IMS_DATA_REQ := f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ExpectedCalleeUri, v_MessageHeader_Invite, p_WaitDuration, -, p_GeolocationInfoAvailable, -, -, p_OIR);

    if (isvalue(v_IMS_DATA_REQ)) {
      v_IMS_DATA_REQ_Value := valueof(v_IMS_DATA_REQ);
      v_InviteRequest := v_IMS_DATA_REQ_Value.Request.Invite;
      v_SDP_Message := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_InviteRequest.messageBody, v_SDP_MessageTemplate, p_GeolocationInfoAvailable);

      // check media description:
      f_IMS_SDP_CheckMediaDescr_AudioCodecs(v_SDP_Message.media_list[0], p_AudioCodecList, v_CheckForbiddenCodecParameters);  /* @sic R5-151968: f_IMS_SDP_CheckMediaDescr_EVS_16000_1, f_IMS_SDP_CheckMediaDescr_AMR_16000_1 replaced by f_IMS_SDP_CheckMediaDescr_AudioCodec sic@
                                                                                                                                 @sic R5-172931: f_IMS_SDP_CheckMediaDescr_AudioCodecs instead of f_IMS_SDP_CheckMediaDescr_AudioCodec sic@
                                                                                                                                 @sic R5-180231: v_CheckForbiddenCodecParameters sic@ */
      v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ_Value, v_SDP_Message);
    }
    return v_InviteRequestWithSdp;
  }


9. 


	Function name
	f_IMS_InviteRequest_MessageHeaderRX

	Reason for change
	Geolocation and GeolocationRouting  not defaulted to omit in the template for Invite Request if condition A8 is false.

	Summary of change
	Geolocation and GeolocationRouting  defaulted to omit in the template for Invite Request if condition A8 is false.

	TTCN module
	IMS_Procedures_CallControl4G5G



	  function f_IMS_InviteRequest_MessageHeaderRX(template (present) SipUrl p_ExpectedCalleeUri,
                                               OptionTag_List p_OptionTagsForSupported := tsc_OptionTagList_100rel,
                                               INVITE_A_2_1_Context_Type p_Context := A_2_1_A4,
                                               boolean p_A3 := pc_MultimediaTelephonyService,    /* @sic R5s130510 additional change 1: tsc_IMS_MTSI replaced by pc_MultimediaTelephonyService sic@ */
                                               boolean p_A8 := false,
                                               boolean p_A11 := false)
    runs on IMS_PTC return template (present) MessageHeader
  {
    var boolean v_IsGiba := f_IMS_PTC_SecurityScheme_IsGiba();  // => A1/A2
    var template (present) MessageHeader v_MessageHeader_Invite := cr_MessageHeader_Dummy;
    var template (present) RouteBody_List v_RouteBodyList;
    var IMS_ProtectedPorts_Type v_Protected;
    var template integer v_Port_us_Contact := *;
    var template integer v_Port_ps_Route := *;
    var template DisplayName v_DisplayName := *;
    var template (present) SipUrl v_SipUrlFrom := ?;    /* per default any SIP URI is accepted in the From header in the first place;
                                                           further checks may be applied in f_IMS_A_2_1_Invite_CommonChecks (depending on the conditions acc. to A.2.1 */
    var template (present) charstring v_Pcscf;
    var charstring v_MediaType := c_sdpApplication;
    var template (present) CallId v_CallId;
    var template (present) integer v_CSeqValue := ?;
    var template (present) From v_From;
    var template (present) To v_To;
    var template AcceptBody_List v_AcceptBodyList := {};
    var OptionTag_List v_OptionTag_List := p_OptionTagsForSupported;
    var boolean v_GRUUsInSIP := pc_IMS_GRUUsInSIP;
    var boolean v_A4 := (p_Context == A_2_1_A4);    // INVITE creating a dialog    @sic R5-166280 sic@
    var boolean v_A5 := (p_Context == A_2_1_A5);    // re-INVITE within a dialog   @sic R5-166280 sic@
    var boolean v_A6 := (p_Context == A_2_1_A6);    // emergency session in case of no registration          @sic R5-166280 sic@
    var boolean v_A19 := (p_Context == A_2_1_A19);  // emergency session in case of no registration (GIBA)   @sic R5-174545 sic@
    var boolean v_A26 := pc_IMS_Session_Timer;      // UE supports Session Timer   @sic R5-192321 sic@
    
    if (pc_IMS_CS_PS_SRVCCMidCall or pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@
      v_OptionTag_List := v_OptionTag_List & tsc_OptionTagList_norefersub;
    }
    if (v_A4 and v_A26) {                           /* @sic R5-192321 sic@
                                                       NOTE: A4, A6, A7 and A19 are mutual exclusive; use of A20, A21 (eCall) is unclear */
      v_OptionTag_List := v_OptionTag_List & tsc_OptionTagList_timer;  /* @sic R5-173024: support of session timer acc. RFC 4028  sic@
                                                                          @sic R5-180230: end of grace period sic@ */
    }
    
    if (v_A6 or v_A19) {  // INVITE for creating an emergency session in case of no registration
      // Route, Contact: ports to be used
      v_Port_us_Contact := ?;
      v_Port_ps_Route := tsc_IMS_PortNumber_5060;
      if (not v_A19) {                                    /* @sic R5-191118 sic@ */
        // From: Display
        v_DisplayName := cr_DisplayName("Anonymous");     /* @sic R5s140005, R5-140901, R5w140111 sic@ */
      }
      v_GRUUsInSIP := false;     /* @sic R5s160366: no GRUU in case of no registration sic@ */
    }
    else {
      // Route, Contact: ports to be used
      if (not v_IsGiba or (p_Context == A_2_1_A7)) {
        v_Protected := f_IMS_PTC_Security_GetProtectedPorts();
        v_Port_us_Contact := v_Protected.Port_us;
        v_Port_ps_Route := v_Protected.Port_ps;
      } else {  /* @sic R5s160366: GIBA sic@ */
        v_Port_us_Contact := ?;
        v_Port_ps_Route := tsc_IMS_PortNumber_5060 ifpresent;
      }
    }

    if (v_A5)  {                  /* re-invite @sic R5s140017 MCC160 implementation sic@ */
      v_RouteBodyList := f_IMS_Request_RouteSet_RouteBodyList_RX();
      v_CallId := f_IMS_PTC_ImsInfo_DialogGetCallId();
      v_From := f_IMS_RequestInDialog_FromHeaderRX();
      v_To := f_IMS_RequestInDialog_ToHeaderRX();
      v_CSeqValue := f_IMS_RequestInDialog_CSeqValueRX();     /* @sic R5s140235, R5s140385 change 3 MCC160 implementation sic@ */
    }
    else {                                       /* normal case: no re-invite */
      // route body acc. to conditions in A.2.1
      v_Pcscf := ?;                              /* @sic R5s140934 change 6: P-CSCF address is checked later on (f_IMS_A_2_1_Invite_CommonChecks -> f_IMS_MessageHeader_Request_CheckRoute) sic@ */
      v_RouteBodyList[0] := cr_RouteBody(*, cr_SipUri_HostPort_lr(v_Pcscf, v_Port_ps_Route));  /* @sic R5s140873: v_Pcscf;  R5s140934 change 6 sic@
                                                                                                  @sic R5-155174: "loose routing" parameter mandatory for any kind of call (i.e. emergency call too, in contrast to R5s150817) sic@ */
      if (not match(p_Context, (A_2_1_A6, A_2_1_A7, A_2_1_A19))) {  // => no emergency call
        v_RouteBodyList[1] := cr_RouteBody(*, cr_SipUri_HostPort_lr(tsc_IMS_Scscf));
      }
      if (p_A8) {
        v_MessageHeader_Invite.geolocation := cr_GeolocationAny;
        v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@
        v_MediaType := tsc_MimeType_MultipartMixed;
      }
      v_CallId := cr_CallId(?);
      v_From := cr_From(v_DisplayName, v_SipUrlFrom);
      v_To := cr_ToDef(p_ExpectedCalleeUri);
    }
    
    if (p_A8) {
      v_MessageHeader_Invite.geolocation := cr_GeolocationAny;
      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingSupported; // @sic R5-134798 sic@
      v_MediaType := tsc_MimeType_MultipartMixed;
    } else {
      v_MessageHeader_Invite.geolocation := omit;
      v_MessageHeader_Invite.geolocationRouting := omit;
    }
    
    v_MessageHeader_Invite.via             := f_IMS_SipRequest_ViaRX();                               /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address
                                                                                                         @sic R5-133358: rport is optional for tcp sic@ */
    v_MessageHeader_Invite.route           := cr_Route(v_RouteBodyList);
    v_MessageHeader_Invite.fromField       := v_From;                                                 /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of FQDN or IP address, check against P-Preferred-Identity header */
    v_MessageHeader_Invite.toField         := v_To;                                                   /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: check of tag */
    v_MessageHeader_Invite.callId          := v_CallId;                                               /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: value acc. to A4 or A5 */
    v_MessageHeader_Invite.cSeq            := cr_CseqDef(v_CSeqValue, "INVITE");                      /* additional checks in f_IMS_A_2_1_Invite_CommonChecks: A5 */
    v_MessageHeader_Invite.supported       := cr_Supported(v_OptionTag_List);
    v_MessageHeader_Invite.contact         := cr_Contact(f_Contact_SipUri_HostPortRX(?, v_Port_us_Contact, v_GRUUsInSIP));  /* additional checks in f_IMS_A_2_1_Invite_CommonChecks:
                                                                                                                               check of FQDN or IP address, check of feature-param for A3, A6, A10, A12
                                                                                                                               @sic R5s150721, R5-153760: cr_SipUri_HostPort replaced by f_Contact_SipUri_HostPortRX sic@
                                                                                                                               @sic R5s160366: v_GRUUsInSIP sic@ */
    v_MessageHeader_Invite.contentType     := cr_ContentType(v_MediaType);
    v_MessageHeader_Invite.maxForwards     := cr_MaxForwardsDef;
    
    // additional headers for specific conditions:
    if (v_IsGiba or v_A6 or v_A19) {
      v_MessageHeader_Invite.securityVerify  := omit;
      if (v_A6 or v_A19) {
        v_MessageHeader_Invite.require       := omit;
        v_MessageHeader_Invite.proxyRequire  := omit;
      }
    } else {   // -> A1 or A7
      v_MessageHeader_Invite.securityVerify  := f_IMS_PTC_BuildSecurityVerifyHeader();
      if (not v_A5) {  /* @sic R5s140017 change 9.3 sic@ */
        v_MessageHeader_Invite.require       := cr_RequireSecAgree;
        v_MessageHeader_Invite.proxyRequire  := cr_ProxyRequireSecAgree;
      }
    }

    if (not v_IsGiba) {
      v_MessageHeader_Invite.pAccessNetworkInfo := cr_PAccessNetworkInfo(f_IMS_PTC_GetRanType());
    }

    if (v_A4) {  /* @sic R5s140017 change 9.3 sic@ */
      v_AcceptBodyList := {cr_AcceptBody(c_sdpApplication), cr_AcceptBody(c_ims3gppApplication)};
    }
    if (pc_IMS_CS_PS_SRVCCMidCall) { // @sic R5-145746 sic@
      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.mid-call+xml"); // add another element in the array
    }
    if (pc_IMS_CS_PS_SRVCCAlert) { // @sic R5-145746 sic@
      v_AcceptBodyList[lengthof(v_AcceptBodyList)] := cr_AcceptBody("application/vnd.3gpp.state-and-event-info+xml"); // add another element in the array
      v_MessageHeader_Invite.recvInfo := cr_RecvInfo_Def(tsc_IMS_InfoPackageType_3gpp);  // @sic R5-145746, R5s150704,  sic@
    }
    if (lengthof(v_AcceptBodyList) > 0) {
      if (v_A5) { // @sic R5s150971 header optional if A5 sic@
        v_MessageHeader_Invite.accept := cr_Accept(superset(all from v_AcceptBodyList)) ifpresent;
      } else {
        v_MessageHeader_Invite.accept := cr_Accept(superset(all from v_AcceptBodyList));
      }
    }
    
    if (p_A8) {
      v_MessageHeader_Invite.geolocation     := cr_GeolocationAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks
      v_MessageHeader_Invite.geolocationRouting := cr_GeolocationRoutingAny;  // Checked in f_IMS_A_2_1_Invite_CommonChecks @sic R5-133151 sic@
    }

    if (p_Context == A_2_1_A7) {
      v_MessageHeader_Invite.pPreferredID := cr_PPreferredID(?);    // Checked in f_IMS_A_2_1_Invite_CommonChecks
    }
    if (p_A3 and v_A4) {                          /* @sic R5s140263 change 5, modified by R5-142949 (-> A4 only) sic@ */
      v_MessageHeader_Invite.pPreferredService := cr_PPreferredService(tsc_IMS_FeatureParamValue_3gpp_icsi_ref);   /* @sic R5s140565 change 2 sic@ */
    }
    if ((p_A3 and v_A4) or p_A11) {               /* @sic R5-142949 sic@ */
      v_MessageHeader_Invite.acceptContact   := cr_AcceptContact;    // Checked in f_IMS_A_2_1_Invite_CommonChecks (depending on conditions)
    }

    if (pc_EarlyMedia and not v_A5) {                                             /* @sic R5-162934: condition A16 for A.2.1  sic@
                                                                                     @sic R5-166280: "not A5" sic@ */
      
      v_MessageHeader_Invite.pEarlyMedia := cr_PEarlyMedia({"supported"});
    }
    return v_MessageHeader_Invite;
  }



